

Doc. 023/1990

*AMORTIZACION Y COSTE DE PRESTAMOS
CON HOJAS DE CALCULO*

ISIDRO SANCHEZ ALVAREZ

AMORTIZACION Y COSTE DE PRESTAMOS CON HOJAS DE CALCULO

Isidro Sánchez Alvarez

Facultad de Económicas

Departamento de Matemáticas

Universidad de Oviedo

Febrero 1990.

	<u>Página</u>
<u>Método de amortización francés</u>	2
Cálculo de los pagos periódicos	2
Obtención directa del cuadro de amortización	3
Macro para crear cuadros de longitud variable	7
Amortización con carencia	9
Coste de la operación de amortización	11
<u>Amortización en progresión aritmética</u>	15
<u>Amortización en progresión geométrica</u>	17
<u>Cuotas constantes de amortización</u>	19
<u>Préstamos subvencionados</u>	20
<u>Intereses anticipados</u>	24
Método alemán	24
Cuotas constantes de amortización	26
Pago único con cobro anticipado de intereses	28
<u>Macro para imprimir</u>	29
Anexo: Cálculo de intereses día a día	31
Anexo: Amortización mediante renta fraccionada	34
Bibliografía	37

Las operaciones de préstamo son una de las aplicaciones tradicionales de las hojas de cálculo en las actividades empresariales. No existe, sin embargo, un estudio pormenorizado de su aplicación a cada uno de los sistemas clásicos de amortización. El objetivo de este trabajo es llevar a cabo un estudio sobre el cálculo del coste y el proceso de amortización por medio de hojas de cálculo, tratando de resolver los principales problemas informáticos que se plantean.

Por otra parte, se hará especial referencia a la nueva normativa del Banco de España sobre información a la clientela por parte de las Entidades de Depósito. Veremos, pues, como se obtiene la Tasa Anual Equivalente (TAE), principal instrumento de comunicación al cliente sobre el coste de la operación.

La hoja de cálculo empleada será la Lotus 123™. Es preciso, por tanto, un dominio de los conocimientos básicos de manejo de dicho programa, así como un conocimiento financiero de la problemática de las operaciones de préstamo, para seguir los sucesivos desarrollos del estudio. Las funciones descritas corresponderán siempre a la versión 2 española de la citada hoja electrónica.

Los sucesivos ejemplos tratan de ser progresivos con la finalidad de facilitar el manejo de la programación.

METODO FRANCES DE AMORTIZACION

CALCULO DE LOS PAGOS PERIODICOS

Este método de amortización se caracteriza por realizar la devolución del capital prestado mediante pagos periódicos iguales (una renta constante).

@AMORT es una función financiera que permite obtener directamente la cuantía de dichos pagos periódicos, conocida la duración del préstamo, el tipo de interés concertado y el capital prestado. El formato de la función es:

@AMORT (principal; interés; periodos)

siendo:

principal = cuantía objeto de préstamo.

interés = interés efectivo por periodo de pago.

periodos = número de plazos (periodos) de amortización.

La fórmula utilizada por esta función para el cálculo del pago a realizar es:

$$\text{pago} = \text{principal} \cdot \frac{\text{interés}}{1 - (1 + \text{interés})^{-\text{periodos}}}$$

Deben observarse algunas reglas financieras respecto a esta función:

1. Los periodos deben introducirse en las mismas unidades de tiempo que las empleadas para el tipo de interés.
2. Los argumentos requeridos para los valores de principal, interés y periodos deben ser necesariamente numéricos, indicándose bien mediante direcciones de celda, nombres de rango de celdas únicas, fórmulas, funciones o simplemente números.

3. El tipo de interés puede introducirse, bien como fracción decimal en tanto por uno (por ejemplo: 0,16), o bien como porcentaje (16%).

A modo de ejemplo, vamos a obtener el importe de los pagos mensuales a realizar durante 4 años para amortizar un préstamo de 1 millón de pts al 13,75% de interés nominal anual. La introducción en una celda de la función:

`@AMORT(1000000;0,1375/12;48)`

nos proporciona la cuantía de 27.201,23 , que corresponde al pago necesario para amortizar dicho principal a un tipo de interés mensual $i_{12} = 0,1375/12 = 0,1145833$ durante 48 meses.

Debe tenerse en cuenta que esta función considera que los pagos se realizan a final del periodo (postpagables) luego para calcular los términos amortizativos equivalentes que venciesen al comienzo del periodo habría que dividir el resultado obtenido por (1 + interés).

OBTENCION DIRECTA DEL CUADRO DE AMORTIZACION

Una primera aproximación a la construcción de tablas de amortización consiste en la copia de las fórmulas que relacionan las distintas magnitudes financieras representadas (cuotas de interés, cuota de amortización, término amortizativo, deuda pendiente, total amortizado) en un rango determinado de celdas.

En la figura 1 se muestra un ejemplo que permite crear cuadros de amortización anual para periodos máximos de 20 años.

Formato

La columna A se ha fijado con un ancho de 7 caracteres, mientras las restantes tienen 12 (la modificación del ancho deberá ajustarse en función de la resolución de la pantalla). Empleando los comandos:

/ Hoja Global Ancho

se fija la anchura global de las columnas en 12 caracteres; después, situándose sobre la columna A, y con el mandato:

/ Hoja Columna Fijar-ancho

se establece la anchura de la primera columna en 7 caracteres. Las celdas D1 y D5 están ajustadas al formato monetario con los comandos

/ Rango Formato Monetario (0 decimales)

Aunque optaremos por operar con cifras redondeadas (práctica actual de los usos bancarios), para utilizar distintos niveles de exactitud basta con exigir un mayor o menor número de decimales.

	A	B	C	D	E
1			Capital	1.000.000 ₺	
2			Interés	13,75%	
3			Años	3	
4			Periodicidad	2	
5			Pagos	208.987 ₺	
6					
7	Plazo	Amortización	Intereses	Pendiente	Pago
8				1000000	
9	1	140237	68750	859763	208987
10	2	149878	59109	709885	208987
11	3	160182	48805	549703	208987
12	4	171195	37792	378508	208987
13	5	182965	26022	195543	208987
14	6	195543	13444	0	208987
15	7	ND	ND	ND	ND
				
248	240	ND	ND	ND	ND

Figura 1

La celda D2 ha sido ajustada a formato de porcentaje con los comandos:

/ Rango Formato % (2 decimales)

El resto de celdas permanecen en el formato general, si bien cabría la opción de

formatear como monetarias las celdas que ocupan los datos numéricos de la tabla.

Rótulos

Los rótulos aparecen con la simple introducción de las palabras en las celdas correspondientes, quedando directamente ajustadas a la izquierda de la celda, tal como aparece en el rango C1..C5. No obstante, los rótulos de la fila 7 han sido centrados con los comandos:

/ Rango Rótulo Centro (A7..E7)

operación que también podría haberse realizado introduciendo el símbolo ^ delante de cada rótulo.

Fórmulas

La celda D5 obtiene el pago periódico mediante la función:

@AMORT (D1;D2/D4;D3*D4)

cifra que se redondea con cero decimales mediante la función matemática correspondiente:

@REDOND((@AMORT(D1;D2/D4;D3*D4));0)

Si bien la presentación en pantalla es la misma sin emplear la función de redondeo (ya que la celda tiene formato monetario de cero decimales), es preciso utilizar esta función dado que la cifra que necesitamos tomar para posteriores operaciones es la cuantía redondeada. En caso contrario, se tomarían también los decimales correspondientes, aunque no apareciesen en pantalla.

La celda D8 contiene la fórmula : +D1 , en cuanto que el capital pendiente al principio de la operación es el importe global del préstamo.

La celda A9 contiene la fórmula : 1+A8 , tomando así el valor 1 y permitiendo, al ser copiada, ir contando el número de plazo.

La celda C9 calcula los intereses redondeados del primer plazo mediante la fórmula:

@REDOND((+D8*\$D\$2/\$D\$4);0)

Recordemos que \$D\$2 y \$D\$4 aparecen con los signos \$ para que dichas celdas de referencia permanezcan fijas al copiar la fórmula en otras celdas, mientras que D8 irá tomando los sucesivos valores de la deuda viva.

Para evitar resultados incoherentes en amortizaciones para

plazos inferiores al máximo, se introduce la fórmula condicional:

`@SI(A9 > D3*D4;@ND;@REDOND((+D8*D2/D4);0))`

que produce el mensaje ND (No Disponible) para los valores posteriores al final del préstamo.

La celda D9, que recoge la deuda pendiente, se obtiene a través de la fórmula: $+D8-B9$, como diferencia entre la deuda viva al principio del periodo y la cuota de amortización correspondiente al mismo periodo.

La celda B9, que recoge la cuota de amortización se obtiene como :

`@SI(A9=D3*D4;D8;-C9+D5)`

es decir, será igual a la diferencia entre el pago y la cuota de interés, excepto en el último periodo, en el cual debe amortizarse obligatoriamente toda la deuda pendiente.

La celda E9 se obtiene como la suma : $+B9+C9$, puesto que colocar directamente el valor obtenido en la celda D5 no tendría en cuenta que el pago correspondiente al último periodo puede ser distinto al inicialmente previsto, como consecuencia de los redondeos realizados.

Una vez que tenemos construida la primera fila del cuadro, basta copiar esta fila en el rango máximo deseado para la tabla. Puesto que queremos construir cuadros para préstamos de duración máxima 20 años y periodicidad no superior a la mensual, deberemos copiar las fórmulas en las 239 filas restantes que como máximo ocuparemos, mediante los comandos:

/ Copiar (desde A9..E9) (en A10..E248)

Entrada de datos

Los datos que es necesario introducir en este ejemplo son:

- el principal del préstamo en la celda D1.
- la tasa de interés nominal anual en la celda D2.
- la duración en años en la celda D3.
- la frecuencia de capitalización en la celda D4.

Ei resto de los datos son obtenidos por la hoja de cálculo de forma que para préstamos de duración inferior al máximo aparecerá el mensaje ND

para aquellas celdas no utilizadas.

Esta forma de crear tablas de amortización, aunque sencilla, presenta sin embargo algunos problemas importantes:

a. Aparecen filas no utilizadas, dada la longitud variable de las operaciones, que hacen poco estética la presentación. Este punto lo resolveremos en el punto siguiente mediante la creación de una macro que permita ajustar la longitud del cuadro de amortización.

b. La impresión supone también la necesidad de crear una macro que permita realizarla rápidamente y ajustándose a la variabilidad de la longitud.

MACRO PARA CREAR CUADROS DE LONGITUD VARIABLE

La necesidad de crear tablas de longitud variable supone exigir a la hoja de cálculo la elección racional del número de filas. A continuación desarrollaremos una macro que permita, sobre el ejemplo anterior, ajustar la longitud de los cuadros de amortización a la deseada, tal como aparece en la figura 2.

	A	B	C	D	E
1			Capital	600.000 ₺	
2			Interés	19,00%	
3			Años	2	
4			Periodicidad	4	
5			Pagos	91.897 ₺	
6					
7	Plazo	Amortización	Intereses	Pendiente	Pago
8				600000	
9	1	63397	28500	536603	91897
10	2	66408	25489	470195	91897
11	3	69563	22334	400632	91897
12	4	72867	19030	327765	91897
13	5	76328	15569	251437	91897
14	6	79954	11943	171483	91897
15	7	83752	8145	87731	91897
16	8	87731	4167	0	91898

Figura 2

Previamente, con el mandato: / Rango Nombre Usar-Rótulos, pondremos el nombre respectivo a las celdas que contienen el capital, interés, periodicidad y pagos. El nombre de la macro: \m , lo introduciremos en la celda I1, de forma que así evitamos que aparezca en la pantalla principal [Debe recordarse que los mandatos de barra y las fórmulas deben ir precedidos de una comilla en las macros].

A partir de la celda J1, en vertical, insertaremos las sentencias que constituyen la macro:

J3: {IR}COMIENZO~
J4: 0~{ABAJO}0~{ARRIBA}/C{ABAJO}~{DCHA}.{DCHA}{DCHA}{DCHA}{DCHA}~
J5: {IR}COMIENZO~
J6: /RB{END}{DCHA}.{END}{ABAJO}~
J7: {IR}RANGO~
J8: +(D3*D4)+8
J9: {EDICION}{CALCULO}{HOME}~~
J10: /CFORMULAS~A10.A
J11:
J12: ~{HOME}

Los nombres de rangos utilizados en la macro se corresponden con las siguientes celdas:

Comienzo = A10

Rango = J11

Formulas = A9..E9

que deberán ser nombradas con el mandato / Rango Nombre Crear y dichos títulos.

Para finalizar, será necesario nombrar la macro con el mandato / Rango Nombre Crear, asignándole la denominación \m . Para ejecutarla bastará con pulsar conjuntamente las teclas: <Alt> m

Cada vez que se modifiquen las variables de entrada (capital, interés, periodicidad, años) será preciso ejecutar la macro para ajustar la longitud de la tabla. Como en este caso no aparecerán filas sobrantes, la celda C9 podremos simplificarla en:

@REDOND((+D8*\$INTERES/\$PERIODICIDAD);0)

AMORTIZACION CON CARENCIA

1. Carencia de amortización

En esta modalidad de amortización, se dispone de un periodo previo con cuotas de amortización nulas, si bien es necesario abonar los intereses al final de cada periodo. El cuadro podremos construirlo a partir del anterior, sin más que modificar las fórmulas de algunas celdas.

En primer lugar, deberemos introducir la nueva variable de entrada, el periodo de carencia, cuyo valor recogeremos en la celda F1.

En la celda B9 se introducirá:

@SI(A9<=\$F\$1;0;@SI(A9=\$AÑOS*\$PERIODICIDAD;D8;-C9+\$PAGOS))

con lo cual se consigue hacer nulas las cuotas de amortización del periodo carencial.

	A	B	C	D	E	F
1			Capital	400.000	∧ Carencia	3
2			Interés	15,00%		
3			Años	3		
4			Periodicidad	3		
5			Pagos	78.807	∧	
6						
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				400000		
9	1	0	20000	400000	20000	
10	2	0	20000	400000	20000	
11	3	0	20000	400000	20000	
12	4	58807	20000	341193	78807	
13	5	61747	17060	279446	78807	
14	6	64835	13972	214611	78807	
15	7	68076	10731	146535	78807	
16	8	71480	7327	75055	78807	
17	9	75055	3753	0	78808	

Figura 3

Debe modificarse a su vez la celda D5, dado que los términos amortizativos serán diferentes a los obtenidos sin carencia. Deberá introducirse: @REDOND((@AMORT(CAPITAL;INTERES/D4;AÑOS*D4-F1));0) fórmula que deduce el número de periodos de carencia del total para obtener el número de periodos en que se produce amortización, obteniendo a partir de éste el nuevo término amortizativo constante.

A modo de ejemplo presentamos la tabla de la figura 3

De la misma forma que en el caso anterior, basta con ejecutar la macro para construir la tabla ajustada en longitud.

2. Carencia de amortización e intereses

En esta variante, el periodo de carencia supone que tanto las cuotas de amortización como las de interés sean nulas. El préstamo deberá, pues, amortizarse en un número de periodos igual al producto:

$$\text{AÑOS} \cdot \text{PERIODICIDAD} - F1$$

como en el caso anterior. Ahora bien, en este caso, el capital a amortizar será el

	A	B	C	D	E	F
1			Capital	500.000 ₺	Carencia	2
2			Interés	12,00%		
3			Años	2		
4			Periodicidad	4		
5			Pagos	97.920 ₺		
6						
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				500000		
9	1	0	0	515000	0	
10	2	0	0	530450	0	
11	3	82006	15914	448444	97920	
12	4	84467	13453	363977	97920	
13	5	87001	10919	276976	97920	
14	6	89611	8309	187365	97920	
15	7	92299	5621	95066	97920	
16	8	95066	2852	0	97918	

Figura 4

capital inicial junto con los intereses acumulados, luego la celda D5 debe modificarse para recoger la nueva cuantía a amortizar introduciendo:

`@REDOND((@AMORT(D1*(1+D2/D4)^F1;D2/D4;D3*D4-F1));0)`

La celda C9 deberá recoger la fórmula:

`@SI(A9<=F1;0;(@REDOND((+D8*$INTERES/$PERIODICIDAD);0)))`

con lo cual los intereses se anulan durante la carencia.

También la celda D9 debe sufrir modificaciones sustanciales al introducirse la expresión:

`@SI(A9<=F1;@REDOND((D8*(1+$INTERES/$PERIODICIDAD));0);+D8-B9)`

que permite recoger fielmente las deudas vivas en cada momento.

A modo de ejemplo, presentamos el cuadro de la figura 4.

COSTE DE LA OPERACION DE AMORTIZACION

La hoja de cálculo Lotus 123 proporciona una función financiera que permite obtener la tasa interna de rentabilidad de un flujo de capitales. Esta función presenta el formato:

`@TIR(valor inicial;rango)`

siendo:

valor inicial = estimación inicial de la tasa de rentabilidad correcta, a partir de la cual comienza el proceso de aproximación.

rango = conjunto de celdas que contienen los flujos de capitales. El primer capital deberá tener signo negativo (desembolso inicial o capital prestado).

Esta función asume que los ingresos procedentes de la inversión se reinvierten a la propia tasa de rentabilidad interna. Para obtener los resultados, la función @TIR utiliza un proceso iterativo de experimentación y error que permite obtener una aproximación adecuada. De ahí la necesidad de un valor inicial a partir del cual iniciar el proceso.

Puesto que las operaciones de préstamo suponen un flujo de capitales, la función @TIR permitirá obtener la tasa interna de la operación. Dicha tasa representará el coste para el prestatario (cuando se tienen en cuenta los flujos

reales del receptor del préstamo) y la rentabilidad para el prestamista (cuando la operación se plantea en términos reales para quien entrega el principal).

El largo proceso iterativo de aproximación hace que la ejecución de la función @TIR en una hoja suponga un alargamiento importante de los procesos. El simple cambio del número de años de amortización puede producir tiempos de ejecución relativamente importantes. Para evitar estos retrasos, se deberá establecer el recálculo de forma manual, lo cual evita el recálculo automático de toda la hoja ante cualquier modificación realizada. Para situarnos en la opción de cálculo manual se ejecuta el mandato:

/ Hoja Global Recalculo Manual

	A	B	C	D	E	F
1			Capital	3.000.000	∧ Carencia	2
2			Interés	17,50%	Comisiones	15000
3			Años	4	Corretajes	9000
4			Periodicidad	2	T.A.E.	18,52%
5			Pagos	663.782	∧ TAE inicial	17,50%
6					Tot. interés	1507696
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				3000000		
9	1	0	262500	3000000	262500	
10	2	0	262500	3000000	262500	
11	3	401282	262500	2598718	663782	
12	4	436394	227388	2162324	663782	
13	5	474579	189203	1687745	663782	
14	6	516104	147678	1171641	663782	
15	7	561263	102519	610378	663782	
16	8	610378	53408	0	663786	

Figura 5

La consideración del coste efectivo nos obliga a incorporar nuevos datos de entrada en la hoja: los gastos originados. A este respecto, sólo consideraremos los gastos más comunes en la práctica financiera: comisiones iniciales (estudio, apertura) y los corretajes. No obstante, cualesquiera otro tipo de

gastos que se desee tener en cuenta se pueden incorporar de forma similar al cálculo del coste.

Tal como aparece en la figura 5, correspondiente a una hoja de trabajo semejante a la de la figura 3 pero completada con el cálculo del coste, en las celdas E2 y E3 introducimos, respectivamente, los rótulos " Comisiones " y "Corretajes", asignando a las celdas F2 y F3 dichos títulos mediante el mandato:

/ Rango Nombre Usar-títulos.

En la celda E4 introduciremos las siglas T.A.E., que hacen referencia a la Tasa Anual Equivalente de coste de la operación de préstamo. Dicha tasa de coste, fijada normativamente por el Banco de España, es la tasa interna del flujo de capitales generados en la operación, excluyendo todos aquellos gastos suplidos por el banco o entidad prestamista. Se trata pues de un coste menor que el realmente soportado por el prestatario, que no incluirá gastos como los corretajes, timbres, gastos notariales, etc

De los dos tipos de gastos iniciales que estamos considerando sólo deberemos tener en cuenta las comisiones de cara a la obtención del TAE. Este valor lo obtendremos en la celda F4 introduciendo la expresión:

$$(1+@TIR(F5/D4;E8..E368))^D4-1$$

expresión que nos transforma directamente el interés por k-ésimo al que resulta la operación en su equivalente anual.

Respecto a este proceso de cálculo, debemos realizar algunas observaciones complementarias:

1. La celda F5 recogerá el tipo de interés nominal anual de partida para llevar a cabo la aproximación del TAE. El valor de esta celda lo introduciremos directamente, teniendo presente que si está muy alejado del valor exacto obtendremos un mensaje de error.

Puesto que tanto la celda F4 como la F5 recogen dos tipos de interés, normalmente expresados en tanto por ciento, daremos el formato de porcentaje a ambas celdas.

2. Dado que los pagos tiene una periodicidad que puede ser distinta de la anual, al obtener el valor de @TIR estaremos calculando un coste

referido siempre al periodo de capitalización. La expresión de la celda F4 transforma esa tasa por k-ésimo de año en su tasa anual equivalente en capitalización compuesta (TAE).

3. El rango comienza en la celda E8, en la cual deberemos colocar un primer capital negativo, que recogerá el capital recibido en préstamo en términos reales: **+COMISIONES-CAPITAL** . Puesto que no nos interesa visualizar el contenido de dicha celda, la ocultaremos con el mandato

/ Rango Formato Oculta

4. El rango termina en la celda E368, lo cual supone que se toma un primer capital negativo (capital prestado menos gastos) y otros 360 capitales (los términos amortizativos). La cantidad de 360 pagos supone un máximo de 30 años con periodos mensuales. Para obtener el coste de préstamos de mayor duración bastará con modificar el rango de la función @TIR extendiéndolo hasta una celda suficientemente alejada. Las operaciones a periodos inferiores ajustan su longitud con la macro anteriormente descrita.

5. Hemos matizado que el coste obtenido en la celda F4 no era el efectivamente soportado por el prestatario, sino el ajustado a la normativa del Banco de España. Para hallar el coste real total, basta con incluir todos los gastos como componentes del coste, incluidos los suplidos. En nuestro caso, supone incluir también el corretaje como gasto inicial, lo cual se lleva a cabo introduciendo en la celda E8 una nueva fórmula:

+COMISIONES+CORRETAJE-CAPITAL

El valor que ahora obtengamos en la celda del TAE será el coste total soportado.

6. En el caso de que apareciesen gastos iniciales diferentes de los que hemos considerado, bastará incluirlos en uno de los dos conceptos (comisiones, corretaje) que hemos utilizado, para poder obtener el coste de la operación.

Si se trata de gastos suplidos por el banco se incluirán en el concepto de corretaje, de forma que afectarán al coste real total pero no al TAE de la operación.

Si se trata de gastos no suplidos por el banco, se incluirán en el concepto de comisiones, de forma que intervendrán tanto en la determinación del coste real total como en el TAE.

Si apareciesen gastos en momentos diferentes del inicial, deberán afectarse los flujos de capitales de los periodos que les corresponda.

La secuencia más rápida para operar con esta hoja de trabajo consiste en:

1. Introducir los datos de entrada deseados.
2. Ejecutar la macro.
3. Pulsar F9 para realizar los cálculos.

Una última información que incorporaremos en esta hoja de trabajo es la suma aritmética de los pagos por intereses. Introduciremos para ello el rótulo "Tot. interes" en la celda E6, y la fórmula

`@SUMA(C9..C368)`

en la celda F6. Debe observarse que este cálculo presenta una limitación similar a la ya enunciada en el punto 4 respecto al rango de la función @TIR.

AMORTIZACION EN PROGRESION ARITMETICA

En esta modalidad de amortización los términos amortizativos varían en progresión aritmética. Para construir la correspondiente tabla, partiremos del ejemplo de la figura 5. En primer lugar, eliminaremos la variable "pago" de C5 para introducir una nueva magnitud: la razón de la progresión. El valor de la razón se introducirá en la celda D5, la cual será preciso renombrar como "razón".

Necesitaremos utilizar una nueva celda (de forma arbitraria hemos elegido B1) para obtener la cuantía del primer pago. En dicha celda introduciremos la fórmula:

$$\frac{(((D1 \cdot D2 / D4) + (D5 \cdot (D3 \cdot D4 - F1))) / (1 - (1 + D2 / D4)^{-(D3 \cdot D4 - F1)})) - (D5 / (D2 / D4)) - (D5 \cdot (D3 \cdot D4 - F1))}{1}$$

expresión que nos permite obtener la cuantía del primer término con cuota positiva de amortización. Esta fórmula incorpora ya la posibilidad de periodos de carencia de amortización. Puesto que este resultado sólo nos interesa como cálculo intermedio, ocultaremos el contenido de la celda mediante el mandato: / Rango Formato Oculta.

La columna E de pagos también deberá modificarse sustancialmente para construir el cuadro. Introduciremos:

`@SI(A9=1+F1;@REDOND(B1;0);@SI(A9=D3*D4;+D8+C9;`

`@SI(A9<1+F1;+C9;+E8+D5)))`

expresión que:

- en el periodo de carencia hace que el pago sea igual a la cuota de interés.
- toma como primer término amortizativo el valor redondeado de la celda B1.
- obtiene los sucesivos pagos sumando la razón al pago anterior
- redondea la última línea de la tabla de amortización.

	A	B	C	D	E	F	
1			Capital	1.000.000	h	Carencia	2
2			Interés		12,00%	Comisiones	5000
3			Años		2	Corretajes	2000
4			Periodicidad		4	T.A.E.	12,97%
5			Razón	10.000	h	TAE inicial	13,00%
6						Tot.interés	172755
7	Plazo	Amortización	Intereses	Pendiente		Pago	
B				1000000			
9	1	0	30000	1000000		30000	
10	2	0	30000	1000000		30000	
11	3	130459	30000	869541		160459	
12	4	144373	26086	725168		170459	
13	5	158704	21755	566464		180459	
14	6	173465	16994	392999		190459	
15	7	188669	11790	204330		200459	
16	8	204330	6130	0		210460	

Figura 6

La última modificación necesaria deberá realizarse en la celda B9, correspondiente a las cuotas de amortización. En esta celda figurará la fórmula: $\text{@SI}(A9<=\$F\$1;0;\text{@SI}(A9=\$AÑOS*\$PERIODICIDAD;D8;-C9+E9))$ que supone la exclusiva modificación de \$PAGOS por E9, dado que en este método el pago será variable.

Los resultados pueden observarse en la figura 6.

Al igual que en las modalidades anteriores, al ejecutar la macro \m se copiarán ajustadamente las fórmulas de la primera línea en el resto del cuadro, que recalcularemos con F9.

De forma semejante podría construirse una tabla genérica con carencia de amortización e interés, siguiendo las mismas restricciones que habíamos considerado en el método francés.

AMORTIZACION EN PROGRESION GEOMETRICA

En esta modalidad, los pagos realizados varían en progresión geométrica cuya razón deberá ser positiva para que la operación tenga sentido. Partiendo de la figura 6, introduciremos algunas modificaciones que nos permitan obtener el nuevo cuadro de amortización.

El primer pago que incluye amortización no nula, y que también recogeremos en la celda B1, se obtiene con la fórmula:

$$\text{@SI}(1+D5=1+D2/D4;D1*(1+D2/D4)/(D3*D4-F1);D1*(1+(D2/D4)-(1+D5))/(1-((1+(D2/D4))^{-(D3*D4-F1)})*(1+D5)^{(D3*D4-F1)}))$$

que obtiene dicha cuantía según la razón sea o no igual a (1+i).

La razón, que recogemos en la celda D5, la expresaremos en

tanto por ciento, luego formatearemos dicha celda como de tipo porcentaje. Esta celda no recoge la razón real de la progresión, sino que ésta la obtiene la hoja de trabajo como la expresión 1+D5. Por tanto, si deseamos que los términos disminuyan podemos introducir un porcentaje negativo en la celda D5, siempre que no supere el 100%, que daría lugar a una razón negativa.

	A	B	C	D	E	F
1			Capital	3.000.000	Carencia	2
2			Interés	15,00%	Comisiones	15000
3			Años	4	Corretajes	9000
4			Periodicidad	2	T.A.E.	15,79%
5			Razón	10,00%	TAE inicial	15,00%
6					Tot.interés	1362492
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				3000000		
9	1	0	225000	3000000	225000	
10	2	0	225000	3000000	225000	
11	3	282088	225000	2717912	507088	
12	4	353954	203843	2363958	557797	
13	5	436280	177297	1927678	613577	
14	6	530359	144576	1397319	674935	
15	7	637630	104799	759689	742429	
16	8	759689	56977	0	816666	

Figura 7

Las columnas de cuotas de amortización, interés y deuda pendiente no necesitan ser modificadas. Tan solo será precisa una variación en la parte final de la fórmula contenida en la celda E9 correspondiente a los pagos. Dicha expresión deberá finalizar:

.... ;@SI(A9<1+\$F\$1;+C9;@REDOND((+E8*(1+\$D\$5));0)))

con lo cual se consigue la variación acumulativa de los pagos.

En la figura 7 puede observarse un ejemplo de amortización.

CUOTAS CONSTANTES DE AMORTIZACION

Este sistema permite construir tablas de amortización con gran facilidad, al permitir obtener directamente la columna de cuotas de amortización.

Partiremos para su elaboración del cuadro de la figura 7.

Prescindiremos, en primer lugar, de las celdas B1, C5 y D5 por representar variables que no utilizaremos.

La celda B9 deberá contener la fórmula:

`@SI(A9<=F1;0;@SI(A9=D3*D4;D8;@REDOND(D1/(D3*D4-F1);0)))`

con lo cual:

- durante la carencia la amortización es nula.
- se redondea la última línea.
- se redondea la cuota constante de amortización.

	A	B	C	D	E	F
1			Capital	1.250.000	Carencia	2
2			Interés	16,00%	Comisiones	6250
3			Años	4	Corretajes	3750
4			Periodicidad	2	T.A.E.	16,90%
5					TAE inicial	16,00%
6					Tot.interés	550000
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				1250000		
9	1	0	100000	1250000	100000	
10	2	0	100000	1250000	100000	
11	3	208333	100000	1041667	308333	
12	4	208333	83333	833334	291666	
13	5	208333	66667	625001	275000	
14	6	208333	50000	416668	258333	
15	7	208333	33333	208335	241666	
16	8	208335	16667	0	225002	

Figura 8

La celda E9 también deberá modificarse introduciendo la expresión: `+B9+C9`, con lo cual los pagos se obtienen como suma aritmética de las cuotas de interés y amortización. Los resultados pueden observarse en la figura 8.

PRESTAMOS SUBVENCIONADOS

Dentro de las ayudas llevadas a cabo por las administraciones Central y Autonómica para potenciar la inversión y la creación de empleo, destacan los préstamos subvencionados. Esta subvención consiste normalmente en una serie de puntos sobre el interés nominal concertado. Los sistemas de amortización de estos préstamos pueden ser cualesquiera de los ya estudiados; la forma de materializar esta subvención admite básicamente dos modalidades:

1. Subvenciones periódicas de los intereses devengados.

En este caso, los términos amortizativos se obtienen en base al tipo de interés inicialmente previsto, si bien la cuota de interés se verá posteriormente reducida en la tasa porcentual de subvención fijada, con lo cual los pagos efectivos a realizar se verán disminuidos en la cuantía de dicha subvención de interés. Con esta modificación, la ley de variación de las anualidades será diferente de la prevista inicialmente.

Para construir la tabla de amortización, partiremos del préstamo de la figura 5, en el cual será preciso llevar a cabo las siguientes modificaciones:

En la celda A1 introduciremos el rótulo "%Subven" para indicar que en la celda B1 se introduzcan los puntos de interés nominales de subvención. Para ello daremos el formato de porcentaje a la celda B1.

En la columna F introduciremos la cuantía de las subvenciones. Introduciremos el rótulo "Subvención" en la celda F7, y la fórmula:

`@REDOND(+D8*B1/$PERIODICIDAD;0)`

en la celda F9, que permitirá obtener la primera cuantía de subvención de intereses.

Puesto que nos interesaba copiar esta fórmula en el resto de la columna F, y con la longitud variable que imponga la duración del préstamo, deberemos modificar el nombre de rango "Fórmulas". Lo llevaremos a cabo mediante el mandato

`/ Rango Nombre Crear (Fórmulas)(rango A9..F9)`

que permite la copia de la fórmula anterior con la longitud deseada en cada ejecución

de la macro.

También será necesario modificar la columna de pago, para lo cual introduciremos en la celda E9 la fórmula: $+B9+C9-F9$ que obtendrá el pago a realizar sumando las cuotas de interés y amortización y deduciendo la subvención correspondiente.

Por último, modificaremos también la forma de cálculo del coste. La subvención no modifica el T.A.E. de la operación financiera, luego podremos obtener dicha tasa con la correspondiente hoja de trabajo. (en este ejemplo, sería la relativa a la figura 5). Sin embargo, sí nos interesa conocer el coste real de la operación, que no coincide con el coste inicialmente previsto menos la deducción porcentual de intereses. Para ello modificaremos el rótulo de la celda E4 introduciendo " Coste real " y en la celda E5 introduciendo " Coste inic. ".

También modificaremos la celda E8, que recordemos ha sido formateada como oculta, introduciendo los gastos por comisiones. Dicha celda tendrá , por tanto, la fórmula: $+F2-D1+F3$

De esta forma, la celda F4 nos permitirá obtener el coste real de la operación, teniendo en cuenta la subvención. Recordemos que la celda F5 se utiliza para introducir un valor inicial de coste para que la hoja comience el proceso iterativo de aproximación al valor real.

Los resultados pueden observarse en la figura 9.

	A	B	C	D	E	F
1	%Subven	.3,50%	Capital	2.000.000	⌘ Carencia	0
2			Interés	13,50%	Comisiones	10000
3			Años	2	Corretajes	6000
4			Periodicidad	2	Coste real	10,97%
5			Pagos	587.127	⌘ Coste inic.	14,50%
6						
7	Plazo	Amortización	Intereses	Pendiente	Pago	Subvención
8				2000000		
9	1	452127	135000	1547873	552127	35000
10	2	482646	104481	1065227	560039	27088
11	3	515224	71903	550003	568486	18641
12	4	550003	37125	0	577503	9625

Figura 9

2. Subvención única e igual al valor actualizado de las subvenciones periódicas de los intereses devengados.

Se trata de una variación del ejemplo anterior en que la subvención financiera de intereses se obtiene como un capital único resultante de actualizar las subvenciones sucesivas en la cuota de interés. Este pago único suele ser entregado en el momento en que se justifican todas las inversiones realizadas, a la entidad prestamista para llevar a cabo una amortización de capital, si bien también puede ser entregada al prestatario.

La tasa de actualización utilizada suele fijarse, de forma más o menos arbitraria, en relación con los intereses de referencia en el mercado financiero, si bien su elección es de gran importancia de cara a la subvención concedida.

La Comunidad Autónoma de Asturias subvenciona sus préstamos según esta modalidad, utilizando un 10% de interés como tasa nominal anual de actualización, no admitiendo plazos de amortización superiores a 6 años.

Partiendo de la figura 4, construiremos, como en los casos anteriores, una hoja que permita obtener la tabla de amortización, el coste del préstamo y hallar la subvención.

En primer lugar, situaremos los nuevos valores de entrada en las celdas iniciales de la hoja. En la celda A1 introduciremos el rótulo " %Subven ", y el valor en puntos de interés nominales se introducirá en la celda B1, la cual formatearemos con rango de porcentaje. En la celda A2 incorporaremos el rótulo " Cuantía " cuyo valor aparecerá en la celda B2 a través de la fórmula que más adelante comentaremos y que recogerá la cuantía única a recibir en concepto de subvención.

En la columna G introduciremos la cuantía de intereses subvencionada. Para ello será preciso escribir en la celda G9 la fórmula:

$$+D8*\$B\$1/\$PERIODICIDAD$$

Puesto que es preciso que esta fórmula se copie, con la longitud variable, en el resto de las filas de la tabla de amortización, será necesario que el rango de nombre que habíamos denominado " Fórmulas " se modifique para incluir

también esta celda. Para ello ejecutaremos el mandato:

/ Rango Nombre Crear (Fórmulas)(rango A9..G9)

con lo cual, cada vez que ejecutemos la macro, esta columna de intereses subvencionados también se ejecutará a una longitud variable y se copiará en el resto de las filas.

El importe de la subvención única lo recogeremos en la celda B2 como el valor actual de todos los importes de intereses subvencionados actualizados al tipo de interés nominal fijado (emplearemos el 10% en nuestros cálculos). Introduciremos pues la fórmula:

@REDOND(@VAN(0,1/\$PERIODICIDAD;G9..G369);0)

que permitirá obtener la subvención para préstamos de duración máxima treinta años y periodicidad mensual. Tal como habíamos comentado al obtener el coste de las operaciones, este rango puede ampliarse para el estudio de amortizaciones a más largo plazo.

	A	B	C	D	E	F
1	%Subven	3,00%	Capital	1.000.000	€ Carencia	3
2	Cuantía	58243	Interés	13,75%	Comisiones	5000
3			Años		3 Corretajes	2000
4			Periodicidad		3 T.A.E.	14,69
5			Pagos	194.400	€ TAE inicial	14,50
6					Tot.interés	303900
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8				1000000		
9	1	0	45833	1000000	45833	
10	2	0	45833	1000000	45833	
11	3	0	45833	1000000	45833	
12	4	148567	45833	851433	194400	
13	5	155376	39024	696057	194400	
14	6	162497	31903	533560	194400	
15	7	169945	24455	363615	194400	
16	8	177734	16666	185881	194400	
17	9	185881	8520	0	194401	

Figura 10

El resultado de estas modificaciones puede observarse en la figura 10. De la misma forma, podría construirse una hoja de cálculo para cualesquiera de las otras modalidades de amortización.

El coste real para el prestatario de la operación no se obtendrá restando al tipo de interés los puntos de subvención, puesto que:

- se trata de una subvención de interés nominal.
- el tipo de interés de actualización influye en el coste.

Para obtener el coste real del préstamo será preciso construir el nuevo cuadro de amortización a que da lugar la subvención y utilizar la función @TIR para hallar el coste implícito que representa.

La principal dificultad estriba en el desconocimiento, en principio, de la fecha en que se hará efectiva la entrega de la subvención.

INTERESES ANTICIPADOS

Dentro de este tipo de préstamos, caracterizados por abonarse los intereses a principios del periodo, estudiaremos tres modalidades:

1. Método Alemán.

Caracterizado por abonarse términos amortizativos constantes, construiremos el cuadro de amortización a partir el gráfico 5.

El pago constante a realizar en esta modalidad se obtiene a través de la expresión:

$$a = C \cdot z / 1 - (1 - z)^n$$

siendo:

a = término amortizativo.

C = capital prestado.

n = número de periodos.

z = tipo de interés anticipado por periodo.

Modificaremos, consecuentemente, la celda D5 para recoger este término, introduciendo la fórmula:

@REDOND((D1*D2/D4)/(1-(1-D2/D4)^(D3*D4-F1));0)

La columna de las cuotas de amortización será la que elegiremos como troncal para la construcción de la tabla; de ahí que sea la que precise mayores modificaciones. Nos basaremos para ello en las relaciones:

$$A_s = A_s + 1 \cdot (1 - z)$$

$$a = A_n$$

Introduciremos en la celda C9 la fórmula:

@REDOND((@SI(A9<=\$F\$1;0;@SI(A9=\$D\$3*\$D\$4;D8;
+\$D\$5*(1-\$D\$2/\$D\$4)^(D\$3*\$D\$4-A9)))));0)

que permite obtener las sucesivas cuotas en función del pago constante calculado en la celda D5.

La columna D, que obtiene las sucesivas deudas pendientes, no sufrirá modificaciones: se obtendrá como diferencia entre la deuda anterior y la cuota amortizativa del periodo (fórmula +D8-B9 en la celda D9). Tampoco será preciso modificar la columna E que recoge los pagos, la cual se obtendrá como suma de amortización e intereses.

Sí deberá modificarse la columna de intereses, dado que en este caso se calcularán sobre la deuda pendiente, no en el periodo anterior, sino en el actual. La fórmula que deberemos introducir en C9 será:

@REDOND(+D9*\$D\$2/\$D\$4;0)

Recordemos que la fila 9 será la que se copiará con longitud variable al ejecutar la macro, luego para recoger los intereses pagados en el primer periodo, que aparecen en la celda C8, será necesario introducir también la fórmula en dicha celda.

La columna E8, que en principio debería recoger el pago realizado por la contraprestación en el momento 0, y que correspondería por tanto a los intereses de la celda C8, aparece en la figura 11 formateada como oculta y recogiendo la fórmula:+C8+COMISIONES-CAPITAL

La razón estriba en la necesidad de utilizar ese campo como desembolso inicial en el cálculo del T.A.E. Si introdujésemos el desembolso inicial

en la celda E7 , y recogiésemos en la celda E8 el pago de intereses, obtendríamos un T.A.E. erróneo puesto que la hoja de cálculo consideraría dicho pago de intereses realizado al final del primer periodo, y sucesivamente el resto de pagos también como realizados un periodo después.

	A	B	C	D	E	F
1			Capital	5.000.000	∧ Carencia	2
2			Interés	15,00%	Comisiones	25000
3			Años	3	Corretajes	15000
4			Periodicidad	2	T.A.E.	17,19
5			Pagos	1.399.745	∧ TAE inic.	17,00
6					Tot.interés	1348982
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8	0		375000	5000000		
9	1	0	375000	5000000	375000	
10	2	0	375000	5000000	375000	
11	3	1107833	291913	3892167	1399746	
12	4	1197657	202088	2694510	1399745	
13	5	1294764	104981	1399746	1399745	
14	6	1399746	0	0	1399746	

Figura 11

Aunque se plantea la dificultad de no aparecer en la columna de pagos el primer desembolso que comprende sólo intereses, hemos optado por la presentación expuesta en la figura 11.

2. Cuotas constantes de amortización.

En este caso, se amortiza en cada periodo la misma cantidad , abonándose los intereses anticipadamente sobre la deuda viva en cada periodo. Así pues, el cuadro será fácil de obtener cubriendo directamente la columna B de cuotas de amortización, obteniendo el resto de valores en función de la misma.

Por similitud con el préstamo de cuotas constantes de amortización y pago de intereses por vencido, partiremos de la tabla de amortización

de la figura 8, en la cual sólo será preciso hacer mínimas modificaciones.

En la celda C9, que recoge la fórmula para el cálculo de intereses, haremos que estos se obtengan, no sobre la deuda pendiente anterior (D8), sino sobre la deuda pendiente durante el periodo de cálculo (D9), con lo cual utilizaremos la expresión:

@REDOND((+D9*\$INTERES/\$PERIODICIDAD);0)

	A	B	C	D	E	F
1			Capital	1.250.000	¶ Carencia	2
2			Interés	15,00%	Comisiones	12500
3			Años	4	Corretajes	3400
4			Periodicidad	2	T.A.E.	17,43%
5					TAE inic.	17,00%
6					Tot.interés	421875
7	Plazo	Amortización	Intereses	Pendiente	Pago	
8			93750	1250000		
9	1	0	93750	1250000	93750	
10	2	0	93750	1250000	93750	
11	3	208333	78125	1041667	286458	
12	4	208333	62500	833334	270833	
13	5	208333	46875	625001	255208	
14	6	208333	31250	416668	239583	
15	7	208333	15625	208335	223958	
16	8	208335	0	0	208335	

Figura 12

Esta fórmula será preciso copiarla en la celda C8 para recoger el pago de intereses que se devenga en el momento de la concesión del préstamo. Al igual que en el caso anterior, la celda E8, que tenemos formateada como oculta, no la vamos a utilizar para recoger el pago de intereses que correspondería al inicio de la operación, puesto que ello nos distorsionaría la posibilidad de obtener el coste. Sin embargo, sí debemos modificar el contenido de la celda E8 introduciendo ese pago de intereses como una menor disposición inicial, a través de la fórmula:

+COMISIONES-CAPITAL+C8

Los resultados pueden observarse en la figura 12.

3. Pago único con cobro anticipado de intereses al tirón.

Una operación de préstamo muy utilizada en la práctica bancaria consiste en que el prestamista descuenta, en el momento de iniciarse la operación, los intereses simples que genera el nominal del préstamo, a cambio de que el prestatario le devuelva el nominal al finalizar la operación.

Tal como aparece en la figura 13, fijaremos el ancho de las columnas en 15 y formatearemos, por una parte las celdas D5, D9, D10, D11 Y D12 como monetarias, y por otra parte, las celdas D6 y D14 como de porcentaje. El capital prestado se introducirá en la celda D5, y el tipo de interés anual en la celda D6. Las celdas D7 y D8 recogen la fecha de inicio y final de la operación. Para introducir las lo haremos incorporando previamente las comillas (") para permitir que sean consideradas como cadenas de caracteres, y por tanto sean aceptadas con dicho formato. Esto nos permitirá transformarlas en el valor numérico correspondiente a través de la función @VALFECHA , y así obtener los intereses. Estos se

	A	B	C	D
1				
2				
3				
4				
5		CAPITAL PRESTADO		500.000 Ft
6		INTERES ANUAL		12,00%
7		FECHA CONTRATO (dd-mmm-aa)		10-OCT-89
8		FECHA DEVOLUCION (dd-mmm-aa)		29-NOV-89
9		COMISIONES		5.000 Ft
10		CORRETAJE		2.300 Ft
11				
12		LIQUIDO RECIBIDO		484.481 Ft
13		INTERESES DESCONTADOS		8.219 Ft
14		T.A.E.		21,28%

Figura 13

recogen en la celda D13 mediante la fórmula:

`@REDOND(+D5*D6*(@VALFECHA(D8)-@VALFECHA(D7))/365;0)`

El líquido efectivamente recibido en la operación se recogerá en la celda D12 a través de la expresión:

`+D5-D13-D9-D10`

Por último, el coste anual equivalente (TAE), que no recogerá los gastos suplidos (corretajes), se obtendrá a través de la fórmula:

`((D5/(D12+D10))^(360/(@VALFECHA(D8)-@VALFECHA(D7))))-1`

MACRO PARA IMPRIMIR

El proceso de impresión de hojas de cálculo es sencillo, pero necesita un alto número de pulsaciones. De ahí que sea uno de los procesos típicamente simplificados a través de macros.

Por otra parte, en el caso de impresión de tablas de amortización, la longitud variable plantea la necesidad de establecer un rango de impresión diferente, que la macro debe tener en cuenta. Debido a que los cuadros de amortización serán normalmente de más de una página, también será preciso especificar condiciones que permitan imprimir las cabeceras de las columnas en todas las páginas.

Para resolver estos problemas, proponemos la macro de la figura 20, que podemos adaptar a cualquiera de las hojas que hemos venido utilizando:

	P	Q
1	v	{IR}Q9~
2	+AÑOS*PERIODICIDAD+8	+P2{CALCULO}{HOME}'~
3		{HOME}
4		/IIEB
5		RA1..F6~
6		AI
7		OBFA7~S
8		RA8..F
9		
10		~IPS

En la celda P2 se establece el cálculo de la última fila que ocupará la tabla de amortización, valor que luego se llevará a la celda Q9 para completar el rango de impresión. Esta operación se realiza en las celdas Q2 y Q3. En una segunda fase, se imprimen los datos de cabecera directamente, eliminando los anteriores bordes (Q4) e introduciendo el rango. La celda Q6 permite reinicializar página (será preciso el previo ajuste de papel en la impresora) para comenzar la impresión. La celda Q7 hace que se repita la fila de cabecera en cada una de las páginas. Las celdas Q8 y Q9 indican el rango de impresión, el cual variará en función del valor que tome Q9, calculado tal como hemos expuesto anteriormente.

Por último, la celda Q10 transmite la orden de imprimir, de ajustar la página al final del proceso y de salir del mandato de impresión.

Recordemos que para que la macro sea ejecutable es preciso nombrarla.

ANEXO I

CALCULO DE INTERESES DIA A DIA

En la práctica bancaria es muy habitual el cálculo de los intereses de cada periodo con base diaria, esto es, obteniendo los intereses correspondientes a los días que median entre dos pagos consecutivos.

En el cuadro de amortización, esto se traduce en variaciones de las cuotas de interés y, como consecuencia, de las cuotas de amortización y de las deudas pendientes. A modo de ejemplo, modificaremos el préstamo de la figura 5 para que recoja esta característica. Los resultados aparecen en la figura 14.

En primer lugar, será preciso introducir nuevos datos en la columna F, que corresponderán a la fecha en que se realizará el pago. Para ello colocaremos el rótulo " Fecha " en la celda F7. Puesto que esta columna recogerá fechas del calendario, le daremos el formato de día con el mandato:

/Rango Formato Día

En la celda F8, introduciremos la fecha de contrato con la fórmula: @FECHA(B4;B3;B2)

Las celdas B4, B3 y B2 recogerán respectivamente el año, mes y día en que comienza la operación. Estos datos de entrada aparecerán como se observa en la figura 14.

El principal problema se plantea para generar las sucesivas fechas correlativas, distantes respectivamente un mes (consideraremos en principio el caso de liquidaciones mensuales). Para ello construiremos una tabla de datos fuera del marco principal. Formatearemos, con un ancho de 3, las columnas que van desde la " S " hasta la " AD ", introduciendo los siguientes valores:

	S	T	U	V	W	X	Y	Z	AA	AB	AC	AD
1	1	2	3	4	5	6	7	8	9	10	11	12
2	31	28	31	30	31	30	31	31	30	31	30	31
3	31	29	31	30	31	30	31	31	30	31	30	31

Este rango S1..AD3 lo nombraremos como TABLA y lo utilizaremos para generar el mismo día del mes siguiente (fecha en que se realiza el próximo pago) sumando los correspondientes valores de la Fila 1 (año no bisiesto) o Fila 2 (año bisiesto) a la fecha de partida.

La fórmula correspondiente a este proceso la introduciremos en la celda F9 para ser posteriormente copiada en el resto de la columna mediante la macro (para llevarlo a cabo modificaremos el rango de nombre FORMULAS para que incluya también la celda F9). La expresión introducida será:

```
@SI((@AÑO(F8)/4)=@ENT(@AÑO(F8)/4);  
F8+@CONSULH(@MES(F8);$TABLA;2);  
F8+@CONSULH(@MES(F8);$TABLA;1))
```

La función @CONSULH permite acceder a la TABLA para consultar el número de días del mes correspondiente a la fecha del pago, diferenciando que el año sea bisiesto o no. Dichos días se suman a la fecha de partida para obtener la nueva fecha de pago, justo un mes después.

La última modificación necesaria corresponderá a la columna de pagos por intereses, que deberán recoger la liquidación día a día. Introduciremos en la celda C9 la fórmula:

```
@SI((@AÑO(F9)/4)=@ENT(@AÑO(F9)/4);  
@REDOND((+D8*$D$2*(F9-F8)/366);0);  
@REDOND((+D8*$D$2*(F9-F8)/365);0))
```

Debe tenerse en cuenta que, para liquidaciones con frecuencia diferente, será necesario modificar los valores de la TABLA para recoger los nuevos periodos de liquidación, o bien establecer tablas alternativas o ampliadas, lo cual complicaría sustancialmente la hoja de trabajo.

	A	B	C	D	E	F
1	Fecha de contrato	Capital	600.000	Carencia		0
2	Día(dd)	20	Interés	14,00%	Comisiones	3000
3	Mes(mm)	8	Años	1	Corretajes	1800
4	Año(aa)	90	Periodicidad	12	T.A.E.	16,04%
5			Pagos	53.872	TAE inic.	14,50%
6					Tot.interés	46503
7	Plazo Amortización	Intereses	Pendiente	Pago	Fecha	
8			600000			20-Ago-90
9	1	46738	7134	553262	53872	20-Sep-90
10	2	47506	6366	505756	53872	20-Oct-90
11	3	47858	6014	457898	53872	20-Nov-90
12	4	48603	5269	409295	53872	20-Dic-90
13	5	49005	4867	360290	53872	20-Ene-91
14	6	49588	4284	310702	53872	20-Feb-91
15	7	50535	3337	260167	53872	20-Mar-91
16	8	50779	3093	209388	53872	20-Abr-91
17	9	51463	2409	157925	53872	20-May-91
18	10	51994	1878	105931	53872	20-Jun-91
19	11	52653	1219	53278	53872	20-Jul-91
20	12	53278	633	0	53911	20-Ago-91

Figura 14

ANEXO II

AMORTIZACION MEDIANTE RENTAS FRACCIONADAS EN PROGRESION

En muchos casos, se plantea la amortización mediante pagos crecientes anualmente en progresión aritmética o geométrica, permaneciendo constantes los desembolsos realizados dentro de cada periodo anual.

Los préstamos para viviendas de protección oficial suelen seguir procesos amortizativos de este tipo. En este caso, los términos crecen un 3% anual y el tipo de interés fijado legalmente corresponde, no a un tipo de interés nominal, sino al TAE efectivamente soportado o recibido por prestamista y prestatario.

A partir del cuadro de la figura 8, podremos construir una nueva tabla que recoja este modelo de amortización, tal como aparece en la figura 15.

El interés de entrada no será el nominal, sino el TAE. Por otra parte, en el caso de viviendas de protección oficial se planteará normalmente un TAE distinto para la entidad bancaria y para el cliente. Incorporaremos, pues, dos nuevas variables de entrada " TAE banco " y " TAE cliente " en las celdas C2..F2. En la celda F3 recogeremos el interés por periodo de pago equivalente al TAE de la operación para el banco. La celda F4 recogerá el mismo concepto pero referido al cliente. Ambas celdas las formatearemos como de tipo porcentaje con 4 decimales. Por último, moveremos las celdas que recogen el total de interés una línea hacia arriba.

Con el mandato:

/ Rango Nombre Usar-Rótulos Derecha

nombraremos las celdas de las columnas D y F con los rótulos de las columnas C y E, respectivamente. Debemos posteriormente suprimir los nombres de celdas que existían en el ejemplo 8 y que se superponen a los creados en la nueva hoja.

Puesto que las cuotas recibidas por el banco no coinciden con las entregadas por el cliente, utilizaremos la columna E para recoger las primeras, y la columna F para las segundas. Encabezaremos dichas columnas con los rótulos " Banco " y " Cliente " centrados, respectivamente. La columna F deberá copiarse

también con longitud variable, luego modificaremos el rango de FORMULAS pasando a ser A9..F9.

Las celdas B1 y B2, que formatearemos como ocultas, las utilizaremos para calcular el primer pago con cuota de amortización positiva para el banco y para el cliente. En la celda B1 introduciremos la fórmula:

```
@SI(1+D5=1+D2;D1*F3/(D2*(D3-F1)*(1+D2)^(-1));  
D1*(1+D2-1-D5)*F3/(D2*(1-((1+D2)^(-D3+F1))*(1+D5)^(D3-F1))))
```

que calcula dicho pago por m-ésimo en función de que se verifique o no que la razón sea igual a 1+interés.

En la celda B2 se obtiene el primer pago amortizativo desde el punto de vista del cliente a través de la expresión:

```
@SI(1+D5=1+F2;D1*F4/(F2*(D3-F1)*(1+F2)^(-1));  
D1*(1+F2-1-D5)*F4/(F2*(1-((1+F2)^(-D3+F1))*(1+D5)^(D3-F1))))
```

que utiliza el mismo criterio.

En la celda B9, que recoge las cuotas de amortización, introduciremos la fórmula:

```
@SI(A9/$D$4<=$F$1;0;-C9+E9)
```

que calcula las cuotas por diferencia entre la cantidad recibida por el banco y la cuota de interés, excepto en el periodo de carencia, en cuyo caso hace nulas las amortizaciones.

En la celda C9, recogeremos la expresión:

```
@REDOND(+D8*$INT.BANCO;0)
```

que obtiene los intereses de cada periodo como el producto de la deuda pendiente por el interés efectivo del periodo.

La celda E9 incorpora la fórmula:

```
@SI(A9<=$F$1*$D$4;C9;@SI(A9=($F$1*$D$4+1);@REDOND($B$1;0);
```

@SI(@ENT(A9/\$D\$4)=(A9-1)/\$D\$4:@REDOND(E8*(1+\$D\$5);0);E8))

la cual permite que:

- el pago durante la carencia sea igual a la cuota de interés.
- el primer pago amortizativo sea el obtenido en la celda B1.
- los términos crezcan año a año.

Por último, la celda F9 recogerá la expresión:

@SI(A9<=\$F\$1*\$D\$4;\$F\$4*D8:@SI(A9=(\$F\$1*\$D\$4+1);@REDOND(\$B\$2;0);

@SI(@ENT(A9/\$D\$4)=(A9-1)/\$D\$4:@REDOND(F8*(1+\$D\$5);0);F8))

que actúa de forma semejante a la fórmula de la celda E9, pero con referencia al tipo de interés (TAE) del cliente.

De forma parecida podría plantearse la amortización en progresión aritmética.

	A	B	C	D	E	F
1			Capital	600.000	Carencia/año	1
2			TAE banco	11,30%	TAE cliente	7,50%
3			Años	4	Int.banco	3,6331%
4			Periodicidad	3	Int.cliente	2,4400%
5			Razón	3,00%	Tot.interés	181069
6						
7	Plazo	Amortización	Intereses	Pendiente	Banco	Cliente
8				600000		
9	1	0	21798	600000	21798	14640
10	2	0	21798	600000	21798	14640
11	3	0	21798	600000	21798	14640
12	4	55383	21798	544617	77181	72958
13	5	57395	19786	487222	77181	72958
14	6	59480	17701	427742	77181	72958
15	7	63956	15540	363786	79496	75147
16	8	66279	13217	297507	79496	75147
17	9	68687	10809	228820	79496	75147
18	10	73568	8313	155252	81881	77401
18	11	76241	5640	79011	81881	77401
20	12	79010	2871	1	81881	77401

Figura 15

BIBLIOGRAFIA:

Adamis, Eddie: Lotus 123. Guía de referencia Anaya-Multimedia - Microsoft de todos los mandatos o comandos, funciones y características. Anaya-Multimedia. Madrid 1988.

Ewing, David P. : Biblioteca de macros para Lotus 123. Anaya-Multimedia. Madrid 1986.

Simpson, Alan: El libro del Lotus 123. Desarrollo de aplicaciones profesionales. Anaya-Multimedia. Madrid 1989.

Trost ,Stanley R. : Lotus 123. Aplicaciones en gestión. Anaya-Multimedia. Madrid 1988.

Doc. 001/1988

JUAN A. VAZQUEZ GARCIA.- Las intervenciones estatales en la minería del carbón.

Doc. 002/1988

CARLOS MONASTERIO ESCUDERO.- Una valoración crítica del nuevo sistema de financiación autonómica.

Doc. 003/1988

ANA ISABEL FERNANDEZ ALVAREZ; RAFAEL GARCIA RODRIGUEZ; JUAN VENTURA VICTORIA.- Análisis del crecimiento sostenible por los distintos sectores empresariales.

Doc. 004/1988

JAVIER SUAREZ PANDIELLO.- Una propuesta para la integración mutijurisdiccional.

Doc 005/1989

LUIS JULIO TASCÓN FERNANDEZ; JOSE MANUEL DIEZ MODINO.- La modernización del sector agrario en la provincia de León.

Doc. 006/1989

JOSE MANUEL PRADO LORENZO.- El principio de gestión continuada: Evolución e implicaciones.

Doc. 007/1989

JAVIER SUAREZ PANDIELLO.- El gasto público del Ayuntamiento de Oviedo (1982-88).

Doc. 008/1989

FELIX LOBO ALEU.- El gasto público en productos industriales para la salud.

Doc. 009/1989

FELIX LOBO ALEU.- La evolución de las patentes sobre medicamentos en los países desarrollados.

Doc. 010/1990

RODOLFO VAZQUEZ CASIELLES.- Investigación de las preferencias del consumidor mediante análisis de conjunto.

Doc. 011/1990

ANTONIO APARICIO PEREZ.- *Infracciones y sanciones en materia tributaria.*

Doc. 012/1990

MONTSERRAT DIAZ FERNANDEZ; CONCEPCION GONZALEZ VEIGA.- *Una aproximación metodológica al estudio de las matemáticas aplicadas a la economía.*

Doc. 013/1990

EQUIPO MECO.- *Medidas de desigualdad: un estudio analítico.*

Doc. 014/1990

JAVIER SUAREZ PANDIELLO.- *Una estimación de las necesidades de gastos para los municipios de menor dimensión.*

Doc. 015/1990

ANTONIO MARTINEZ ARIAS.- *Auditoría de la información financiera.*

Doc. 016/1990

MONTSERRAT DIAZ FERNANDEZ.- *La población como variable endógena.*

Doc. 017/1990

JAVIER SUAREZ PANDIELLO.- *La redistribución local en los países de nuestro entorno.*

Doc. 018/1990

RODOLFO GUTIERREZ PALACIOS; JOSE MARIA GARCIA BLANCO.- *"Los aspectos invisibles" del declive económico: el caso de Asturias.*

Doc. 019/1990

RODOLFO VAZQUEZ CASIELLES; JUAN TRESPALACIOS GUTIERREZ.- *La política de precios en los establecimientos detallistas.*

Doc. 020/1990

CANDIDO PAÑEDA FERNANDEZ.- La demarcación de la economía (seguida de un apéndice sobre su relación con la Estructura Económica).

Doc. 021/1990

JOAQUIN LORENCES.- Margen precio-coste variable medio y poder de monopolio.

Doc. 022/1990

MANUEL LAFUENTE ROBLEDO; ISIDRO SANCHEZ ALVAREZ.- El T.A.E. de las operaciones bancarias.

Doc. 023/1990

ISIDRO SANCHEZ ALVAREZ.- Amortización y coste de préstamos con hojas de cálculo.

Doc. 024/1990

LUIS JULIO TASCÓN FERNANDEZ; JEAN-MARC BUIGUES.- Un ejemplo de política municipal: precios y salarios en la ciudad de León (1613-1813).

Doc. 025/1990

MYRIAM GARCIA OLALLA.- Utilidad de la teoría de las opciones para la administración financiera de la empresa